A Moon-Centered, Inertial Coordinate System for Spice

This memo describes the steps used to create a right-handed, orthogonal selenocentric or moon-centered inertial frame definition and data file ("frame kernel") for Spice. The fundamental plane of this inertial system is the lunar mean equator and the fundamental x-axis is the IAU node of J2000. The y-axis is advanced 90 degrees along the lunar equator from the x-axis, and the z-axis is perpendicular to the mean equator of the moon. The term mean indicates that precession has been accounted for, but not the effect of nutation.

The following figure illustrates the orientation of this coordinate system relative to the Earth's mean equator and north pole of J2000 (EME2000). The x-axis or Q-vector is formed from the cross product of the Earth's mean pole of J2000 and the Moon's north pole relative to EME2000.

In general, the x-axis is called the IAU node of epoch. For this application, we will set or "freeze" the epoch to J2000 (January 1.5, 2000).



Figure 1. Moon mean equator and IAU node of epoch coordinate system

Lunar pole right ascension and declination relative to EME2000

The following two equations describe the time evolution of the right ascension and declination of the moon's mean pole, in degrees, with respect to the Earth mean equator and equinox of J2000 (EME2000) coordinate system.

$$\begin{aligned}\alpha = {} & 269.9949 + 0.0031T - 3.8787\sin E1 - 0.1204\sin E2 \\ & + 0.0700\sin E3 - 0.0172\sin E4 + 0.0072\sin E6 \\ & - 0.0052\sin E10 + 0.0043\sin E13\end{aligned}$$

$$\begin{aligned}\delta = {} & 66.5392 + 0.0130T + 1.5419\cos E1 + 0.0239\cos E2 \\ & - 0.0278\cos E3 + 0.0068\cos E4 - 0.0029\cos E6 \\ & + 0.0009\cos E7 + 0.0008\cos E10 - 0.0009\cos E13\end{aligned}$$

In these equations, T is the time in Julian centuries given by

$$T = \left(JD - 2451545.0\right)/36525$$

and JD is the TDB (Barycentric Dynamical Time) Julian Date.

The trigonometric arguments, in degrees, for the pole orientation equations are

$$\begin{aligned}E1 &= 125.045 - 0.0529921d \\ E2 &= 250.089 - 0.1059842d \\ E3 &= 260.008 + 13.0120009d \\ E4 &= 176.625 + 13.3407154d \\ E6 &= 311.589 + 26.4057084d \\ E7 &= 134.963 + 13.0649930d \\ E10 &= 15.134 - 0.1589763d \\ E13 &= 25.053 + 12.9590088d\end{aligned}$$

where $d = JD - 2451545$ is the number of days since January 1.5, 2000. These equations are given in "Report of the IAU/IAG Working Group on Cartographic Coordinates and Rotational Elements of the Planets and Satellites: 2000", *Celestial Mechanics and Dynamical Astronomy*, **82**: 83-110, 2002.

The three rotation angles used to create an Euler frame definition file for Spice are given by

$$\theta_1 = -90^\circ - \alpha$$

$$\theta_2 = -90^\circ + \delta$$

$$\theta_3 = 0$$

In order to define a frame relative to the IAU node of J2000, the values of both *T* and *d* used in the polynomials evaluated to compute these angles are identically zero.

Spice frame definition

The following are the contents of a simple ASCII text file named `moon_j2000.tf` that provides the frame definition information needed by Spice.

```
\begintext

 lunar mean equator and IAU node of J2000 frame definition

\begindata

 FRAME_MOON_J2000          =  4902
 FRAME_4902_NAME           = 'MOON_J2000'
 FRAME_4902_CLASS          =  5
 FRAME_4902_CLASS_ID       =  4902
 FRAME_4902_CENTER         =  301
 FRAME_4902_RELATIVE       = 'J2000'
 FRAME_4902_DEF_STYLE      = 'PARAMETERIZED'
 FRAME_4902_FAMILY         = 'EULER'
 FRAME_4902_EPOCH          =  @2000-JAN-1/12:00:00
 FRAME_4902_AXES           =  ( 3  1  3 )
 FRAME_4902_UNITS          = 'DEGREES'
 FRAME_4902_ANGLE_1_COEFFS = ( -356.857733   0.0   0.0 )
 FRAME_4902_ANGLE_2_COEFFS = ( -24.3588973   0.0   0.0 )
 FRAME_4902_ANGLE_3_COEFFS = (    0.0        0.0   0.0 )
```

Additional information can be found in the “FRAMES Required Reading” document, NAIF Document No. 349.10, last modified 27 NOV 2006.

The numerical elements of the EME2000-to-Moon mean equator and IAU node of J2000 transformation matrix are as follows.

```
  0.998496505205088      -5.481540926807404E-2    0.000000000000000
  4.993572939853833E-2    0.909610125238044       0.412451018902688
 -2.260867140418499E-2   -0.411830900942612       0.910979778593430
```

The following is the Fortran subroutine that was used to compute the two angles defined in the Spice frame definition file. For the example described in this document, the input argument `xjdate` was equal to `2451545`.

```
      subroutine mm2000 (xjdate, tmatrix)

c     eme2000 to moon mean equator and IAU node
c     of epoch transformation matrix

c     input

c      xjdate = TDB julian date

c     output
```

```
c       tmatrix = transformation matrix

c      Orbital Mechanics with Fortran

c      ************************************

       implicit double precision (a-h, o-z)

       dimension phat_moon(3), rmoon(3), vmoon(3)

       dimension xvec(3), xhat(3), yvec(3), yhat(3)

       dimension zhat(3), hv(3), hhat(3), tmatrix(3, 3)

c      conversion factor – degrees to radians

       data dtr /1.745329251994330d-2/

c      time arguments

       t = (xjdate - 2451545.0d0) / 36525.0d0

       d = xjdate - 2451545.0d0

c      iau 2000 pole orientation

       e1 = 125.045d0 -  0.0529921d0 * d

       e2 = 250.089d0 -  0.1059842d0 * d

       e3 = 260.008d0 + 13.0120009d0 * d

       e4 = 176.625d0 + 13.3407154d0 * d

       e5 = 357.529d0 +  0.9856003d0 * d

       e6 = 311.589d0 + 26.4057084d0 * d

       e7 = 134.963d0 + 13.0649930d0 * d

       e8 = 276.617d0 +  0.3287146d0 * d

       e10 = 15.134d0 -  0.1589763d0 * d

       e13 = 25.053d0 + 12.9590088d0 * d

       rasc_pole = 269.9949d0 + 0.0031d0 * t
     &             - 3.8787d0 * sin(dtr * e1)
     &             - 0.1204d0 * sin(dtr * e2)
     &             + 0.0700d0 * sin(dtr * e3)
     &             - 0.0172d0 * sin(dtr * e4)
     &             + 0.0072d0 * sin(dtr * e6)
     &             - 0.0052d0 * sin(dtr * e10)
     &             + 0.0043d0 * sin(dtr * e13)

       decl_pole = 66.5392d0 + 0.0130d0 * t
     &             + 1.5419d0 * cos(dtr * e1)
```

```
     &            + 0.0239d0 * cos(dtr * e2)
     &            - 0.0278d0 * cos(dtr * e3)
     &            + 0.0068d0 * cos(dtr * e4)
     &            - 0.0029d0 * cos(dtr * e6)
     &            + 0.0009d0 * cos(dtr * e7)
     &            + 0.0008d0 * cos(dtr * e10)
     &            - 0.0009d0 * cos(dtr * e13)

c     compute the unit vector in the direction of the moon's pole

      phat_moon(1) = cos(rasc_pole * dtr) * cos(decl_pole * dtr)

      phat_moon(2) = sin(rasc_pole * dtr) * cos(decl_pole * dtr)

      phat_moon(3) = sin(decl_pole * dtr)

c     define eme2000 z-axis unit vector

      zhat(1) = 0.0d0
      zhat(2) = 0.0d0
      zhat(3) = 1.0d0

c     compute x-direction (IAU node of epoch)

      call vcross (zhat, phat_moon, xvec)

      call uvector(xvec, xhat)

c     compute y-direction

      call vcross (phat_moon, xhat, yvec)

      call uvector(yvec, yhat)

c     load elements of transformation matrix

      tmatrix(1, 1) = xhat(1)
      tmatrix(1, 2) = xhat(2)
      tmatrix(1, 3) = xhat(3)

      tmatrix(2, 1) = yhat(1)
      tmatrix(2, 2) = yhat(2)
      tmatrix(2, 3) = yhat(3)

      tmatrix(3, 1) = phat_moon(1)
      tmatrix(3, 2) = phat_moon(2)
      tmatrix(3, 3) = phat_moon(3)

      return
      end
```

The following figure illustrates the geometry of the EME2000 coordinate system. Both coordinate system illustrations used in this memo were extracted from JPL D-32296, “Lunar Constants and Models Document” dated September 23, 2005.



Figure 2. Earth mean equator and equinox of J2000 coordinate system